

DC-16 Output Expansion Board (#31216)

High-Current Output Expansion Board

General Description


Whether you build props or design small-scale industrial controls, there will likely come a time when you need to control high-current DC devices like relays and solenoids. The DC-16 is designed just for this purpose, giving the engineer the ability to control up to 16 DC loads (per board). Outputs are controlled remotely through a simple serial interface that allows the host microcontroller to control up to four DC-16 relay boards.

Applications

- Props and holiday displays
- Small-scale industrial control

Features

- Controls up to 16 12- to 24-volt, high-current outputs (through ULN2803 drivers)
- Serial control for connection to host microcontroller
- User-configured baud rate (2400 or 38.4k) for serial control
- User-configured address allows up to four DC-16 boards on one serial connection
- Compact size: 2.6" x 4.2"



Mounting the DC-16

Mount the DC-16 above a non-conductive surface using ½" (12 mm) stand-offs and appropriate hardware. Mounting the DC-16 above a conductive (metal) surface could lead to the possibility of a hazardous condition.

Power Connection / Power Switch

Power to the DC-16 is provided through a 2.1 mm, center-positive barrel connector (commonly used with "wall-wart" power supplies).




CAUTION: Before connecting power to the DC-16 ensure that it is not resting on a metallic (conductive) surface; doing so could damage the DC-16 circuitry and external power supply.

NOTE: Do not exceed 30 volts at the input of the DC-16. Doing so will cause the onboard regulator to shut-down and prevent operation of the controller. When in doubt, use a regulated power supply.


A three-position switch controls power on the DC-16. In position 0, all power is off. In position 1, power is supplied to the regulator and to the onboard command processor. In position 1, the DC-16 is active for program testing via the serial link only. In position 2, the voltage supplied at the power connector is routed to the V+ terminal for use by external devices (relays, solenoids, lights, etc.).

Connections – Switched DC Outputs

The output connectors (OUT1 – OUT16) provide a connection to the ground side of the power supply when the output is active, therefore devices controlled by the DC-16 should be connected between the V+ and respective OUTx terminal. Remember that power will not be available at the V+ terminal unless the power switch is set to position 2. Schematically, a typical connection would look like this:



Advanced designers may elect to use an external supply for the switched devices. If, for example, you would like to run the DC-16 from a 12-volt supply and switch solenoids using a separate 24-volt supply, you must connect the GND (ground) terminal to the ground side of the external supply; DO NOT connect any part of the external supply to the DC-16's V+ terminal. When using an external supply make the connections as shown below.




Note the addition of the 1N4001 suppression diode when using the external supply. This is recommended because the diodes internal to the ULN2803 are not connected to the external supply when using this configuration.



NOTE: Please consult the ULN2803 documentation (available online) before designing with the DC-16 to ensure that the ULN2803 is capable of sinking the amount of current required by the switched devices you intend to use.

Connections – Serial Control

For control of its port outputs the DC-16 is be connected to a host like the Prop-1 controller or a BASIC Stamp 2 series microcontroller in a Board-of-Education (BoE). To make the connection between the host controller and the DC-16 use a 3-wire servo extender cable (e.g., Parallax #805-00001 or #805-00002). When making the connection it is important to note the color-coded polarity of the extender cable: W = white, R = red, B = black.



Baud Rate Selection

The DC-16 supports two baud rates: 2400 baud for the Prop-1 (BS1 family) and 38.4k baud for the BS2-family and other microcontrollers. Baud rate is set with a jumper on the **Baud** header.

Jumper	Baud Rate
Out	2400 (Prop-1, BS1, and compatible controllers)
In	38.4k (BS2 and compatible controllers)

Address Selection

Using a serial connection allows the host controller to connect to up to four DC-16 boards; to do this, however, each must have a unique address. The DC-16 board address is set with jumpers on the **A0** and **A1** headers.

A1	A0	Address
Out	Out	0 (%00)
Out	In	1 (%01)
In	Out	2 (%10)
In	In	3 (%11)

DC-16 Serial Protocol

The DC-16 serial connection conforms to the Parallax AppMod bi-directional communications protocol, which operates at TTL (0 – 5v) using “open-true” baud mode to enable daisy-chaining devices. All exchanges are initiated by the host controller. The format of host communications to the DC-16 is as follows:

```
"!DC16", address, command {, data {, data }}
```

... where “!DC16” is the preamble that allows the DC-16 to exist on the same communications line as other devices (e.g., Parallax Servo Controller), *address* is the board address set with jumpers on A0 and A1, *command* is a single character command designator, and *data* is any value that may be required by command. The following section details valid commands (“V”, “S”, “X”, “L”, “H”, “P”, and “G”) for the DC-16.

“V” – Get DC-16 Firmware Version

Syntax: “!DC16”, *address*, “V”

Reply: 3-byte (ASCII) version string

Prop-1 Example:

```
SEROUT Sio, OT2400, ("!DC16", %11, "V")
SERIN Sio, OT2400, id0, id1, id2
DEBUG "DC-16 Version ", #@id0, #@id1, #@id2, CR
```

“S” – Setup All Outputs

Syntax: “!DC16”, *address*, “S”, *statusLo*, *statusHi*
 – *statusLo* controls OUT1..OUT8, *statusHi* controls OUT9..OUT16
 – 1 bit = On, 0 bit = Off

Reply: none

Prop-1 Example:

```
SEROUT Sio, OT2400, ("!DC16", %11, "S", %00001111, %00000000)
```

In this example outputs OUT1..OUT4 will be activated, all other outputs are deactivated.

“X” – Clear All Outputs

Syntax: “!DC16”, *address*, “X”

Reply: none

Prop-1 Example:

```
SEROUT Sio, OT2400, ("!DC16", %11, "X")
```

In this example all outputs (OUT1..OUT16) are deactivated.

"L" – Setup "Low" Outputs (OUT1..OUT8)

Syntax: `"!DC16", address, "L", status`
– *status* controls OUT1..OUT8
– 1 bit = On, 0 bit = Off

Reply: none

Prop-1 Example:

```
SEROUT Sio, OT2400, ("!DC16", %11, "L", %11100000)
```

In this example outputs OUT6..OUT8 will be activated, outputs OUT1..OUT5 deactivated, while outputs OUT9..OUT16 are unaffected.

"H" – Setup "High" Outputs (OUT9..OUT16)

Syntax: `"!DC16", address, "H", status`
– *status* controls OUT9..OUT16
– 1 bit = On, 0 bit = Off

Reply: none

Prop-1 Example:

```
SEROUT Sio, OT2400, ("!DC16", %11, "H", %00000011)
```

In this example outputs OUT9..OUT10 will be activated, outputs OUT11...OUT16 deactivated, while outputs OUT1..OUT8 are unaffected.

"P" – Set/Reset Individual Output Port

Syntax: `"!DC16", address, "P", portNum, newState`
– *portNum* is output: 1..16
– *newState* is new state of relay: 1 = On, 0 = Off

Reply: none

Prop-1 Example:

```
SEROUT Sio, OT2400, ("!DC16", %11, "P", 7, 1)
```

In this example, output #7 (OUT7) is activated.

"G" – Get Ports Status

Syntax: `"!DC16", address, "G"`

Reply: Two byte value (sent *Little Endian*) that holds current relay status
– Bit0 = OUT1, Bit1 = OUT2, ... Bit15 = OUT16
– 1 indicates port is On; 0 indicates relay is Off.

Prop-1 Example:

```
Main:
SEROUT Sio, OT2400, ("!DC16", %11, "G")
SERIN  Sio, OT2400, statLo, statHi
```

In this example `statLo` will hold the current state of OUT1..OUT8, `statHi` will hold the state of OUT9..OUT16.

Example Program

The following program demonstrates the DC-16 features under serial control using the Prop-1 (BS1) controller. When using the Prop-1 the serial baud rate is set to 2400 (Baud jumper removed).

```
' =====
'
'   File..... DC16_Test.BS1
'   Purpose.... DC-16 Features Test
'   Author..... Team EFX -- Copyright (C) 2005 Parallax, Inc.
'   E-mail..... teamefx@parallax.com
'   Started....
'   Updated.... 07 FEB 2006
'
'   {$STAMP BS1}
'   {$PBASIC 1.0}
'
' =====
'
' -----[ I/O Definitions ]-----
'
SYMBOL  Sio                = 0
'
' -----[ Constants ]-----
'
SYMBOL  Baud                = OT2400
SYMBOL  Addr                = %11          ' %00 - %11
'
' -----[ Variables ]-----
'
SYMBOL  id0                 = B0
SYMBOL  id1                 = B1
SYMBOL  id2                 = B2
```

```

SYMBOL  dcStat      = W2      ' status bits
SYMBOL  statLo     = B4      ' low byte of dcStat
SYMBOL  statHi     = B5      ' high byte of dcStat

' -----[ Initialization ]-----

Reset:
  DEBUG CLS

' -----[ Program Code ]-----

Main:
  ' get version
  SEROUT Sio, Baud, ("!DC16", Addr, "V")
  SERIN  Sio, Baud, id0, id1, id2
  DEBUG "DC-16 Version ", #@id0, #@id1, #@id2, CR

Test_Set:
  SEROUT Sio, Baud, ("!DC16", Addr, "S", %00001111, %00000000)
  GOSUB Show_Status
  PAUSE 500
  SEROUT Sio, Baud, ("!DC16", Addr, "L", %11110000)
  GOSUB Show_Status
  PAUSE 500
  SEROUT Sio, Baud, ("!DC16", Addr, "S", %00000000, %00001111)
  GOSUB Show_Status
  PAUSE 500
  SEROUT Sio, Baud, ("!DC16", Addr, "H", %11110000)
  GOSUB Show_Status
  PAUSE 500

Test_Reset:
  ' all off
  SEROUT Sio, Baud, ("!DC16", Addr, "X")
  GOSUB Show_Status
  PAUSE 500

Test_PortCtrl:
  FOR id0 = 1 TO 16
    SEROUT Sio, Baud, ("!DC16", Addr, "P", id0, 1)
    GOSUB Show_Status
    PAUSE 100
  NEXT
  FOR id0 = 1 TO 16
    SEROUT Sio, Baud, ("!DC16", Addr, "P", id0, 0)
    GOSUB Show_Status
    PAUSE 100
  NEXT

```

```
Test_Ignore:
  ' send invalid port number
  SEROUT Sio, Baud, ("!DC16", Addr, "P", 17, 1)
  GOSUB Show_Status

  END

' -----[ Subroutines ]-----

Show_Status:
  SEROUT Sio, Baud, ("!DC16", Addr, "G")
  SERIN  Sio, Baud, statLo, statHi
  DEBUG CR, "Status = ", #%dcStat
  RETURN RETURN
```

For additional programming examples be sure to visit us on the web:

www.parallax.com
www.parallaxfx.com
forums.parallax.com